


# Isabelle/Spartan—A Dependent Type Theory Framework for Isabelle

Joshua Chen 

University of Innsbruck, Austria

---

## Abstract

This paper introduces Isabelle/Spartan, an implementation of intensional dependent type theory with cumulative universes as an object logic in the Isabelle proof assistant. In contrast to other systems supporting dependent type theory, Isabelle is based on simple type theory—yet I show how its existing logical framework infrastructure is able to handle automation tasks that are typically implemented on the source code level of dependently-typed systems. I also go some way in integrating the propositions-as-types paradigm with the declarative Isar proof language. Isabelle/Spartan supports book HoTT and the univalence axiom, and its capabilities are demonstrated by the formalization of foundational results from the Homotopy Type Theory book.

**2012 ACM Subject Classification** Theory of computation → Logic and verification; Theory of computation → Type theory; Theory of computation → Higher order logic

**Keywords and phrases** Proof assistants, Logical frameworks, Dependent type theory, Homotopy type theory

**Category** System description

**Funding** Partially supported by ERC project “SMART”, starting grant no. 714034.

## 1 Introduction

Proof assistants based on dependent type theory have historically been built “from the ground up” to support their single logical foundation. In contrast to such systems are logical frameworks [11], which are designed to allow a user to work with a wide range of different object logics as environments for formalization and proof, at the potential cost of having less specific automation for any particular formalism. More recently, there has been work towards creating new logical frameworks explicitly designed to support dependent type theories as object logics [1, 5]. All of these systems are themselves dependently-typed.

In contrast, Isabelle [14, 16] is a simply-typed proof assistant and logical framework. Of its multiple object logics Isabelle/HOL is arguably the most well-known, however many other logics have been created since Isabelle’s inception and are still bundled along with its distribution [9]. Among these is an early logic by Paulson [8] based on extensional Martin-Löf type theory, which has not been further developed. In light of considerable recent progress in the field, it seems appropriate to revive support for dependent type theory in Isabelle.

## Motivation

The potential benefits of support for dependent type theory in Isabelle to both the proof assistant and type theory communities seem attractive. One such benefit is the possibility of encoding other versions of dependent type theory, enabling one to rapidly experiment with different formulations and prove meta-theoretic results about them. Support for dependent type theory in a simply-typed setting will also pave the way for greater compatibility between proof assistant libraries, allowing for the porting of results between systems of different formalisms. More ambitiously, the preliminary work presented here in translating between a dependent type theoretic formalism and the Isar language suggests the possibility

(momentarily ignoring concerns of consistency) of HOL-based and dependently-typed “sub-logics” coexisting under one object logic, allowing the user to choose whichever formulation best suits their particular development.

## Contributions

In this paper I introduce *Isabelle/Spartan* (Spartan), a new object logic based on dependent type theory<sup>1</sup>.

In the first half of this paper I present an encoding of intensional dependent type theory with cumulative Russell-style universes in the *Isabelle/Pure* (Pure) meta-logic, and discuss issues and design decisions that arise. Due to small but significant differences between the semantics of Pure and the formalism of Martin-Löf-style type theories, a naïve translation of the latter into the former preserves neither adequacy nor soundness. The rules of Spartan have been formulated to avoid the most obvious sources of this failure; however, as of this writing, soundness and completeness of the encoding with respect to its intended semantics have not yet been proved.

In the second half, I give an overview of the implementation of the system, showing how the existing Isabelle infrastructure may be used to solve tasks—such as typechecking and term elaboration—that are typically considered specific to dependently-typed systems and implemented as routines in their source code. Progress is also made in integrating the propositions-as-types paradigm with the declarative Isar proof language [17]. Although the system currently lacks some automation, one can already easily state and prove nontrivial results from the Homotopy Type Theory book [13] in a style mostly familiar to users of dependently-typed proof assistants.

## Source code

The work presented in this paper has been implemented as a library of standard ML and Isabelle theory files. References to specific files are given as footnotes throughout. The source code is available at <https://github.com/jaycech3n/Isabelle-Spartan>.


## Related work

One of the earliest object logics for Isabelle was Paulson’s *Isabelle/CTT* (CTT) [8] for computational type theory, in the footsteps of which the present work follows. Indeed, the ideas of implementing typechecking as a tactic and of tackling subgoals in order of decreasing “rigidity” already appear in CTT. On the other hand, as CTT implements extensional Martin-Löf type theory without universes it requires less automation for equality reasoning and universe levels than Spartan, and is also unsuited to the homotopical viewpoint.

Andromeda [5] is a new LCF-style proof assistant explicitly designed to support the formulation and subsequent use of user-defined type theories. It provides a general-purpose meta-language to define constructors, types and inference rules, as well as a small trusted nucleus that checks well-typedness and derivability of statements. It is, however, more restrictive than Isabelle in the kinds of type theory it is able to specify as it only allows the formulation of inference rules using the four judgment types of traditional Martin-Löf type theory. In particular, the interval object of cubical type theory is not amenable to this

---

<sup>1</sup> The name is inspired by talks given by Bauer [4] on minimal type theories that support univalence and the homotopical interpretation.



■ **Figure 1** Inference rules of Isabelle/Pure.

framework. In this respect the Isabelle-based work presented here is more flexible, though one would need to consider issues of adequacy, soundness and completeness of the encodings.

Dedukti [1, 3] is a logical framework based on the  $\lambda\Pi$ -calculus Modulo, which is also able to encode a wide variety of logics including dependent type theory. It is, however, designed as a proof checker and is hence not geared towards the interactive proof assistant style of working.

Finally, one cannot speak of logical frameworks without mentioning the LF family of languages, in particular the ELF and TWELF [10, 12] systems. These differ from the work here in that they are focused more on enabling one to specify and prove properties *about* object logics instead of letting one work *inside* them, while the vision for Spartan is to allow one to do both within a single system.

## 2 The Isabelle Logical Framework

To provide context for the presentation of Isabelle/Spartan, this section briefly introduces the functionality of the Isabelle logical framework as well as the Isabelle/Pure meta-logic. Pure is a minimal logic designed primarily to enable users to encode and work with the terms, formulas and inference rules of object logics in a natural deduction style. It is based on rank-one polymorphic simple type theory with a base type *prop* to represent logical propositions, together with three constants

$$\Rightarrow :: \text{prop} \rightarrow \text{prop} \rightarrow \text{prop}, \quad \bigwedge :: (\alpha \rightarrow \text{prop}) \rightarrow \text{prop}, \quad \equiv :: \alpha \rightarrow \alpha \rightarrow \text{prop}$$

expressing implication, universal quantification and equality. As is usual, let us write the Church-style universal quantifier  $\bigwedge(\lambda x. P)$  as  $\bigwedge x. P$ . The rules governing the logical constants are shown in Figure 1. Derivability, discharge of assumptions, and side condition checking are handled by the Isabelle system itself. As a simple type theory, Pure also has lambda terms built in, enabling us to encode terms, types and judgments using higher-order abstract syntax. Alpha-conversion as well as beta- and eta-reduction is automatically handled by the system, and the Isabelle simplifier proves substitution rules for terms. In addition, the framework provides functionality to declare new base types and constants, and to state axioms and inference rules. More details can be found in the Isabelle documentation.

### 111 **3 The Logic of Isabelle/Spartan**

112 This section details the basic setup of the Isabelle/Spartan object logic.<sup>2</sup>

#### 113 **3.1 Semantics**

114 The intended semantics of Spartan is a minimal dependent type theory consisting of the  $\Pi$ ,  
 115  $\Sigma$  and identity types, along with a countable cumulative hierarchy of Russell-style universe  
 116 types. This theory itself is standard, and mostly follows the development given in the  
 117 appendix of the Homotopy Type Theory book [13]. Work is ongoing to establish adequacy,  
 118 soundness and completeness theorems for the encoding presented here, but until then—as  
 119 noted by Paulson [8]—the object logic is best justified by directly considering the meaning  
 120 explanations of the rules.

#### 121 **3.2 Judgments**

122 We begin by declaring a meta-type  $o$  for the class of terms and types of the object logic.  
 123 We then declare a constructor `has_type` ::  $o \rightarrow o \rightarrow \text{prop}$ , written as infix  $(:)$ , to encode  
 124 the typing judgment. Since we have chosen to work with Russell-style universes, types are  
 125 themselves terms and must have the same meta-type. Working with Tarski-style universes  
 126 would allow us to maintain the type/term distinction and formulate the typing judgment  
 127 constant as `has_type` ::  $i \rightarrow t \rightarrow \text{prop}$ , at the cost of having to introduce interpretation  
 128 operators everywhere. Judgmental equality is shallowly embedded via the built-in Pure  
 129 equality ( $\equiv$ ), which forgets type information but allows us to easily reuse the Isabelle simplifier  
 130 to compute terms.

131 Here there is a subtle but important difference between the theory and its implementation.  
 132 In theory, all judgments  $\Gamma \vdash t : T$  are entailed by an explicit context of typings, which  
 133 automatically ensures that all statements only contain well-typed terms. In contrast, Spartan’s  
 134 variable contexts are encoded as implications in the Pure logic, which means that it is possible  
 135 to form formulas  $t : T$  containing untypable terms. However, these formulas will not (pending  
 136 soundness) be provable.

#### 137 **3.3 Universes**

138 To implement universe types we first axiomatize a hierarchy of levels isomorphic to the  
 139 standard natural numbers with their usual order. We declare a meta-type  $lvl$  and the  
 140 constants `0`, `S` and `<` for the zero level, successor level and the order relation. Universes are  
 141 then formed by a single constructor `U` ::  $lvl \rightarrow o$ . Figure 2 shows the rules governing levels  
 142 and universes.

#### 143 **3.4 Types and Terms**

144 The constants for small types, their constructors and their eliminators are formulated using  
 145 Church-style semantics and listed in Figure 3. Type families as well as function arguments to  
 146 dependent eliminators are encoded using meta- instead of object-lambda terms. For example,  
 147 in theoretical presentations the  $\Sigma$ -eliminator is given by a term

148 `SigInd(A, B, C, f, p)`

---

<sup>2</sup> Source code: `Spartan.thy`.

```

axiomatization
  0 :: lvl
  S :: lvl → lvl
  lt :: lvl → lvl → prop                (infix <)
  U :: lvl → o
  where
  O_min: 0 < S(i)
  lt_S: i < S(i)
  lt_trans: i < j ⇒ j < k ⇒ i < k
  U_hierarchy: i < j ⇒ U(i): U(j)
  U_cumulative: A: U(i) ⇒ i < j ⇒ A: U(j)

```

■ **Figure 2** Universe types.

```

axiomatization
  Pi :: o → (o → o) → o
  lam :: o → (o → o) → o
  app :: o → o → o                    (infix `)

  Sig :: o → (o → o) → o
  pair :: o → o → o                    (<_,_>)
  SigInd :: [o, o → o, o → o, o → o → o, o] → o

  Id :: [o, o, o] → o
  refl :: o → o
  IdInd :: [o, [o, o, o] → o, o → o, o, o, o] → o

```

■ **Figure 3** Type and term constructors.

whose third and fourth arguments are meant to be, respectively, a type family  $C: (\sum_{x:A} B(x)) \rightarrow U_i$  and a function  $f: \prod_{x:A, y:B(x)} C(x, y)$  inductively defining the value of  $C$  on all  $p: \sum_{x:A} B(x)$ . In the implementation, these have to be given as the simply-typed meta-functions  $C :: o \rightarrow o$  and  $f :: o \rightarrow o \rightarrow o$ . However, after the  $\Pi$  type has been axiomatized Isabelle's coercive subtyping functionality (Section 12.3 of [15]) is used to coerce object functions into meta functions, which allows users to ignore this distinction most of the time.

Isabelle syntax translations convert between the notations  $\prod x: A. B$ ,  $\sum x: A. B$ ,  $\lambda x: A. b$  and  $x =_A y$  to the internal representations  $\text{Pi}(A, \lambda x. B)$ ,  $\text{Sig}(A, \lambda x. B)$ ,  $\text{lam}(A, \lambda x. b)$  and  $\text{Id}(A, x, y)$ . The types  $\prod x: A. B$  and  $\sum x: A. B$  are abbreviated to  $A \rightarrow B$  and  $A \times B$  when  $B$  is a constant type family.

### 3.5 Inference Rules

Following [7], we define a translation **enc** from the judgments of dependent type theory into the Pure logic, by sending

$$x_1: A_1, \dots, x_n: A_n \vdash \mathcal{I}$$

to the universally-quantified Pure implication

$$\bigwedge x_1, \dots, x_n. \llbracket x_1: A_1; \dots; x_n: A_n \rrbracket \implies \mathbf{enc}(\mathcal{I}),$$

where

$$\mathbf{enc}(t: T) := t: T, \quad \mathbf{enc}(a \equiv b: T) := a \equiv b$$

is the encoding of the typing and equality assertions previously discussed (Section 3.2). This translation is recursively extended to inference rules by defining

$$\mathbf{enc}\left(\frac{\mathcal{J}_1 \quad \dots \quad \mathcal{J}_k}{\mathcal{J}}\right) := (\llbracket \mathbf{enc}(\mathcal{J}_1); \dots; \mathbf{enc}(\mathcal{J}_k) \rrbracket \implies \mathbf{enc}(\mathcal{J})).$$

Note that entailment and derivability are both translated to Pure implication, and that the order of variable typing assumptions is forgotten. Since we use the built-in Pure equality the only rules that need to be axiomatized for judgmental equality are  $\Pi$ - and  $\Sigma$ -congruence. The full list of logical rules is given in Figures 4 and 5. In the implementation, these rules are further organized into three named theorem collections **intros**, **elims** and **comps** for introduction, elimination and computation rules, facilitating their use by proof methods.

One needs to take care of the particular formulation of the rules one encodes, as a naïve translation into the Pure logic easily breaks adequacy (and by extension, soundness). This can be seen by the following example. Consider the standard formulation of the  $\Pi$ -introduction rule

$$\frac{\Gamma, x: A \vdash b: B}{\Gamma \vdash \lambda x: A. b: \prod x: A. B}$$

which would be encoded as

$$(\bigwedge x. x: A \implies b(x): B(x)) \implies \lambda x: A. b(x): \prod x: A. B(x)$$

according to the translation defined above. Taking this rule as an axiom would let us conclude the nonsense statement

$$\lambda x: \langle a, b \rangle. \mathcal{U}(0): \prod x: \langle a, b \rangle. \mathcal{U}(\mathcal{S}(0))$$

axiomatization where

$$\begin{aligned}
\text{PiF: } & \llbracket \bigwedge x. x: A \implies B(x): \mathbb{U}(i); A: \mathbb{U}(i) \rrbracket \implies \prod x: A. B(x): \mathbb{U}(i) \\
\text{PiI: } & \llbracket \bigwedge x. x: A \implies b(x): B(x); A: \mathbb{U}(i) \rrbracket \implies \lambda x: A. b(x): \prod x: A. B(x) \\
\text{PiE: } & \llbracket f: \prod x: A. B(x); a: A \rrbracket \implies f \text{ ` } a: B(a) \\
\text{beta: } & \llbracket a: A; \bigwedge x. x: A \implies b(x): B(x) \rrbracket \implies \lambda x: A. b(x) \equiv b(a) \\
\text{eta: } & f: \prod x: A. B(x) \implies \lambda x: A. (f \text{ ` } x) \equiv f \\
\text{Pi\_cong: } & \llbracket A: \mathbb{U}(i); \\
& \quad \bigwedge x. x: A \implies B(x): \mathbb{U}(i); \\
& \quad \bigwedge x. x: A \implies B'(x): \mathbb{U}(i); \\
& \quad \bigwedge x. x: A \implies B(x) \equiv B'(x) \rrbracket \\
& \implies \prod x: A. B(x) \equiv \prod x: A. B'(x) \\
\\
\text{SigF: } & \llbracket \bigwedge x. x: A \implies B(x): \mathbb{U}(i); A: \mathbb{U}(i) \rrbracket \implies \sum x: A. B(x): \mathbb{U}(i) \\
\text{SigI: } & \llbracket \bigwedge x. x: A \implies B(x): \mathbb{U}(i); a: A; b: B(a) \rrbracket \implies \langle a, b \rangle: \sum x: A. B(x) \\
\text{SigE: } & \llbracket p: \sum x: A. B(x); \\
& \quad A: \mathbb{U}(i); \\
& \quad \bigwedge x. x: A \implies B(x): \mathbb{U}(i); \\
& \quad \bigwedge p. p: \sum x: A. B(x) \implies C(p): \mathbb{U}(i); \\
& \quad \bigwedge x y. \llbracket x: A; y: B(x) \rrbracket \implies f(x, y): C(\langle x, y \rangle) \rrbracket \\
& \implies \text{SigInd}(A, B, C, f, p): C(p) \\
\text{Sig\_comp: } & \llbracket a: A; \\
& \quad b: B(a); \\
& \quad \bigwedge x. x: A \implies B(x): \mathbb{U}(i); \\
& \quad \bigwedge p. p: \sum x: A. B(x) \implies C(p): \mathbb{U}(i); \\
& \quad \bigwedge x y. \llbracket x: A; y: B(x) \rrbracket \implies f(x, y): C(\langle x, y \rangle) \rrbracket \\
& \implies \text{SigInd}(A, B, C, f, \langle a, b \rangle) \equiv f(a, b) \\
\text{Sig\_cong: } & \llbracket \bigwedge x. x: A \implies B(x) \equiv B'(x); \\
& \quad A: \mathbb{U}(i); \\
& \quad \bigwedge x. x: A \implies B(x): \mathbb{U}(i); \\
& \quad \bigwedge x. x: A \implies B'(x): \mathbb{U}(i) \rrbracket \\
& \implies \sum x: A. B(x) \equiv \sum x: A. B'(x)
\end{aligned}$$

■ **Figure 4** Rules of Isabelle/Spartan:  $\Pi$  and  $\Sigma$  types.

```

axiomatization where
  IdF:   $\llbracket A : \mathsf{U}(i); a : A; b : A \rrbracket \Longrightarrow a =_A b : \mathsf{U}(i)$ 
  IdI:   $a : A \Longrightarrow \mathsf{refl}(a) : a =_A a$ 
  IdE:   $\llbracket p : a =_A b;$ 
         $a : A;$ 
         $b : A;$ 
         $\bigwedge x y p. \llbracket p : x =_A y; x : A; y : A \rrbracket \Longrightarrow C(x, y, p) : \mathsf{U}(i);$ 
         $\bigwedge x. x : A \Longrightarrow f(x) : C(x, x, \mathsf{refl}(x)) \rrbracket$ 
         $\Longrightarrow \mathsf{IdInd}(A, C, f, a, b, p) : C(a, b, p)$ 
  Id_comp:  $\llbracket a : A;$ 
             $\bigwedge x y p. \llbracket p : x =_A y; x : A; y : A \rrbracket \Longrightarrow C(x, y, p) : \mathsf{U}(i);$ 
             $\bigwedge x. x : A \Longrightarrow f(x) : C(x, x, \mathsf{refl}(x)) \rrbracket$ 
             $\Longrightarrow \mathsf{IdInd}(A, C, f, a, a, \mathsf{refl}(a)) \equiv f(a)$ 

```

■ **Figure 5** Rules of Isabelle/Spartan: Identity type.

186 due to the semantics of the material implication, and the fact that the encoding does not *a*  
 187 *priori* rule out ill-typed terms.<sup>3</sup> In order to prevent such occurrences, we add in typing *side*  
 188 *conditions* to the premises of the inference rules, requiring the types and terms appearing  
 189 in them to be well-typed. In the example above this entails adding the premise  $A : \mathsf{U}(i)$  to  
 190 obtain the rule `Pil`. This potentially creates more proof obligations for the user, but is mostly  
 191 mitigated by automation.

## 192 4 The Isabelle/Spartan Implementation

### 193 4.1 Theorems and Proofs

194 Theorem goals in Spartan are stated using the Isar `schematic_goal` command instead of the  
 195 usual `theorem` or `lemma` commands. This allows us to have schematic variables—Isabelle’s  
 196 notion of holes—in the goal state in order to support proof term synthesis, implicit arguments  
 197 and term elaboration. In addition, the object logic provides schematic versions `theorem*`,  
 198 `lemma*` and `corollary*` of the usual Isar goal commands.<sup>4</sup>

199 Proofs can be written in a mixture of tactic-style “apply”-proof scripts and declarative  
 200 Isar, modulo a few technical issues. Most significantly, implicit arguments (Section 4.5),  
 201 which are implemented using schematic variables, are forbidden from appearing in context  
 202 assumptions declared using the `assume/assumes` Isar commands. Thus most of the time one  
 203 would work using tactic-style proof scripts.

204 Isabelle provides a way to structure such scripts via the `subgoal` command. This, however,  
 205 fixes any schematic variables in the goal, turning them into free variables, which interferes

<sup>3</sup> One underlying reason for this is our use of Russell-style universes, which forces the object terms and types to have the same syntactic category.

<sup>4</sup> Source code: `theorem_keywords.ML`.



■ **Table 1** Basic methods for Isabelle/Spartan.

Method	Description
<b>typechk</b>	Solves rigid typing goals.
<b>rule</b>	Backwards reasoning: resolves the conclusion of a goal with the conclusion of a rule. Supports reasoning with types as propositions.
<b>intro, intros</b>	Refines the proof term by applying appropriate introduction rules (single and multi-step versions).
<b>reduce</b>	Computes terms; may fail if a term has not yet been sufficiently elaborated.
<b>equality</b>	Automates induction on identity types.

with the synthesis of proof terms. Instead the object logic provides a new **focus** command<sup>5</sup> which focuses on the topmost subgoal in a proof while leaving schematic variables untouched, allowing us to synthesize specific subterms of the proof. The full syntax is

`focus [premises [⟨hyps⟩]] [vars ⟨var⟩+]`

which also optionally moves goal premises out into the context and renames the fixed variables. The Coq-inspired bullet variations `»,` `■`, `►` and `~` are also provided to help further organize proofs.

Spartan extends the basic methods of Isabelle/Pure with the methods shown in Table 1.<sup>6</sup> These are set up to support the Isar “fixes–assumes–shows” style of goal statement, as opposed to writing the entire goal as a single Pure proposition. This parallels the distinction between the local context of hypotheses and the goal proper in dependently-typed systems.

## 4.2 Typechecking

The **typechk** method forms a key component of the automation provided by the logic, and is hooked into the other methods in Table 1 in order to automatically discharge the additional typing side conditions discussed in Section 3.5, which keeps them mostly hidden from the user. It is also installed as an additional simp-solver, which allows the Isabelle simplifier to handle typing conditions that arise in the course of simplifying terms. The method itself is implemented as a tactic that performs proof search using the type introduction and elimination rules, as well as any rules declared with the attribute **typechk**. However, it will only solve *rigid* typing goals, which are statements  $t : T$  where the head of  $t$  is not a schematic variable. This prevents it from wrongly instantiating schematic variables, which is especially important as it is also used to infer implicit arguments for term elaboration (Section 4.5).

## 4.3 Methods

The **rule** method performs backward resolution of the goal conclusion with given theorems. It extends the functionality provided by the Pure rule method in that it can also use rules whose propositional content is encoded as a type. That is to say, for example, that the result of the method invocations

```
apply (rule <⋀ x y. x : A ⟹ y : B(x) ⟹ f(x,y) : C(x,y)>)
```

<sup>5</sup> Source code: `focus.ML`.

<sup>6</sup> Source code: `tactics.ML`, `equality.ML`.

236 and

```
237
238 apply (rule < $f: \prod x: A. \prod y: B(x). C(x, y)$ >)
239
```

240 on the goal

```
241
242 goal :
243    $?prf: C(a, b)$ 
244
```

245 is the same—the schematic variable  $?prf$  is refined to  $f(?a, ?b)$ , subject to the new subgoals  
246  $?a: A$  and  $?b: B(?a)$ .

247 The **intro** and **intros** methods are used to refine proof terms; they are essentially **rule**  
248 restricted to resolving with the type introduction rules.

249 The **reduce** method performs simplification of terms by invoking the simplifier with  
250 type computation rules and other judgmental equalities declared **comp**. The method will  
251 occasionally fail if the term being reduced has not been sufficiently elaborated, which typically  
252 means that the typechecker is refusing to guess instantiations of implicit arguments.

## 253 4.4 Equality

254 Integrating the propositions-as-types approach to equality reasoning with the Isar language  
255 is slightly involved. This is best illustrated with an example. Suppose we wish to prove the  
256 transitivity of equality, which is naturally stated in Isar as shown in Listing 1.

■ **Listing 1** Transitivity of equality in Isar.

```

lemma * Id_transitive :
  assumes
     $A: \mathcal{U}(i)$ 
     $x: A$ 
     $y: A$ 
     $z: A$ 
     $p: x =_A y$ 
     $q: y =_A z$ 
  shows
     $?prf: x =_A z$ 
257
```

258 In a dependently-typed foundation, this statement corresponds to a judgment

259  $x, y z: A, p: x = y \vdash ?prf: \prod_{q: y = z} x = z$

260 whose proof is straightforward: induct on  $p$  to reduce the goal to inhabiting

261  $\prod_{q: x = z} x = z,$

262 which is trivial. In Isar, however, the assumption  $q: y = z$  is “free-floating” in the context,  
263 and we first have to “push it in” to the type of the conclusion to form the requisite dependent  
264 product before we can apply the identity elimination rule, after which point we want to “pull  
265 the new assumption”  $q: x = z$  back out into the context. The **equality** method automates  
266 such translations between the dependently-typed and Isar formulations.

## 267 4.5 Term Elaboration

268 Spartan currently implements a simple semi-automated form of term elaboration, which  
269 works as follows. Holes for implicitly inferred arguments can be inserted into definitions using

the syntax `?`, or directly into goal statements using `{}`. These are expanded into schematic variables by an automatic syntax phase translation<sup>7</sup> and hidden from the user with further syntax translations, only becoming visible when needed in proof goal obligations. Much of the time, however, many of these obligations are automatically solved by the typechecker tactic which is run after every call to a method provided by Spartan. Hence the user does not even see these obligations because they are automatically solved by the typechecker, which fills the hole with the inferred argument.

## 5 Practical Examples

In its current form, Isabelle/Spartan is already able to formalize nontrivial results from the Homotopy Type Theory book [13]. The following examples provide a demonstration of its capabilities and highlight opportunities for improvement.

### 5.1 Path Lifting

To demonstrate equality reasoning and term elaboration we prove the path lifting property.<sup>8</sup> This states that given a type family  $P$  over  $A$ ,  $x, y: A$ ,  $p: x = y$  and  $u: P(x)$ , we have that  $(x, u) = (y, \text{transport}^P(p, u))$ .

```

lemma* pathlift:
  assumes
    A: U(i)
     $\bigwedge x. x: A \implies P(x): U(i)$ 
    x: A
    y: A
    p: x =A y
    u: P(x)
  shows
    {} : <x, u> = <y, trans(P, p, u)>

```

The initial goal state is

```

goal:
  ?*1: <x, u> = <y, trans(P, p, u)>

```

Schematic variables beginning with `?*` arise from holes; in this case the one for the proof term. To begin, we apply induction on  $p$ :

```

apply (equality <p: _>)

```

The expression `<p: _>` is a direct *fact reference* that pattern matches to the context assumption  $p: x =_A y$ . Now the goal state reads

```

goal:
   $\bigwedge x u. \llbracket x: A; u: P(x) \rrbracket \implies ?b(x, u): <x, u> = <x, \text{trans}(P, \text{refl}(x), u)>$ 

```

The typechecker has taken care of all the other side conditions, leaving us to focus on the core of the proof. The subexpression `trans(P, refl(x), u)` normalizes to  $u$ . Having proved this result earlier and declared it as a `comp` rule, we then

<sup>7</sup> Source code: `implicits.ML`.

<sup>8</sup> Source code: `Identity.thy`.

```

316
317 apply reduce

```

to further refine the goal, leaving

```

320
321 goal :
322  $\bigwedge x u. \llbracket x : A; u : P(x) \rrbracket \implies ?b(x, u) : \langle x, u \rangle = \langle x, u \rangle$ 
323

```

This is finished off with

```

325
326 apply intro
327 done
328

```

The Isabelle output panel now displays the synthesized proof term. By switching off implicit notation with the command

```

331
332 no_translations
333  $x = y \leftarrow x =_A y$ 
334  $\text{trans}(P, p) \leftarrow \text{CONST transport}(A, P, x, y, p)$ 
335

```

we see that the  $\Sigma$  type for the equality as well as the endpoints  $x$  and  $y$  of  $p$  have been automatically inferred.

## 5.2 Bi- and Quasi-Inverses

Let us prove that bi-inverses are quasi-inverses. The proof will use almost all the features of Isabelle/Spartan that have been developed so far (with the exception of equality induction and universe level reasoning).<sup>9</sup>

The statement is

```

343
344 lemma* biinv_imp_qinv:
345 assumes
346  $A : \mathcal{U}(i)$ 
347  $B : \mathcal{U}(i)$ 
348  $f : A \rightarrow B$ 
349 shows
350  $\{ \} : \text{biinv}(f) \rightarrow \text{qinv}(f)$ 
351

```

where **biinv** and **qinv** have been defined earlier in the usual way.

We begin with

```

354
355 apply intro
356 unfolding biinv_def
357

```

to pull the premise out into the context and unfold its definition. The goal now reads

```

359
360 goal :
361  $\bigwedge u. u : (\sum g : B \rightarrow A. g \circ f \sim \text{id}_A) \times (\sum g : B \rightarrow A. f \circ g \sim \text{id}_B) \implies ?b(u) : \text{qinv}(f)$ 
362

```

Next we have to use the Pure **erule** method, together with an explicit call to typecheck side conditions,

```

365
366 apply (erule SigE, typechk)+
367

```

in order to split the sum in the premise into its components to obtain

---

<sup>9</sup> Source code: `Equivalence.thy`.

```

369 goal:
370    $\bigwedge u\ u'\ y\ g\ y'\ g'\ y''. \llbracket g: B \rightarrow A; \ y': g \circ f \sim \text{id}_A; \ g': B \rightarrow A; \ y'': f \circ g' \sim \text{id}_B \rrbracket$ 
371    $\implies ?f(u, u', y, g, y', g', y'') : \text{qinv}(f)$ 
372
373

```

374 The universally-quantified variables  $u', y$  that have appeared come from the unused names in  
 375 the nondependent function type of  $g, g': B \rightarrow A$ , and can be safely ignored. The basic Pure  
 376 method `erule` performs *elim-resolution* which matches on the term  $u$  we want to eliminate  
 377 from the premise, but does not automatically handle side conditions.

378 In order to continue we will need to explicitly refer to the newly-separated components.  
 379 We use the `focus` command to name them and pull their properties out of the goal statement  
 380 and into the Isar context:

```

381 focus premises vars _ _ _ g H1 h H2
382
383

```

384 obtaining

```

385 goal:
386    $?f(u\_ , u'\_ , y\_ , g, H1, h, H2) : \text{qinv}(f)$ 
387
388

```

389 The universally-quantified variables have now been fixed, and Isabelle automatically invents  
 390 names for the variables we left unspecified. The properties  $g, h: B \rightarrow A$ ,  $H1: g \circ f \sim \text{id}_A$  and  
 391  $H2: f \circ h \sim \text{id}_B$  can now be explicitly referenced.

392 Now we can prove that  $f$  is a quasi-inverse. First we unfold the definition of `qinv` and  
 393 refine it with

```

394 unfolding qinv_def
395 apply intro
396
397

```

398 to obtain the two proof obligations

```

399 goal (2 subgoals):
400   1.  $?a(u\_ , u'\_ , y\_ , g, H1, h, H2) : B \rightarrow A$ 
401   2.  $?b(u\_ , u'\_ , y\_ , g, H1, h, H2) :$ 
402       $(?a(u\_ , u'\_ , y\_ , g, H1, h, H2) \circ f \sim \text{id}_A) \times (f \circ ?a(u\_ , u'\_ , y\_ , g, H1, h, H2) \sim \text{id}_B)$ 
403
404

```

405 It helps the proof structure here to use focus bullets. Thus

```

406 ■ by (rule <g: _>)
407
408

```

409 finishes the first subgoal and instantiates the schematic variable  $?a$  with  $g$ , leaving

```

410 goal:
411    $?b(u\_ , u'\_ , y\_ , g, H1, h, H2) : (g \circ f \sim \text{id}_A) \times (f \circ g \sim \text{id}_B)$ 
412
413

```

414 We refine the conjunction and prove the first conjunct with

```

415 ■ apply intro
416   apply (rule <H1: _>)
417
418

```

419 Finally, it only remains to prove the second conjunct. This is done by constructing a  
 420 sequence of homotopies ultimately showing that  $g \sim h$ , from which, together with  $H2: f \circ h \sim$   
 421  $\text{id}_B$ , the result follows. This is best done in a “calculation” proof block, the outline of which  
 422 is

```

423 proof -
424   have  $?a: g \sim g \circ f \circ h$ 
425

```

```

426   <proof>
427   moreover have ? $\beta$ :  $g \circ f \circ h \sim h$ 
428   <proof>
429   ultimately have ? $\gamma$ :  $g \sim h$ 
430   <proof>
431   then have ? $\delta$ :  $f \circ g \sim f \circ h$ 
432   <proof>
433   thus {}:  $f \circ g \sim \text{id}_B$ 
434   <proof>
435   qed

```

437 The details of the subproofs are not very enlightening and are hence omitted. They mostly  
 438 consist of rewriting with the basic Pure **subst** method and properties of homotopy, together  
 439 with typechecking. Having proved this final subgoal we obtain the fully-elaborated proof  
 440 term, which is displayed in the output panel.

## 441 6 Conclusion and Future Work

442 The implementation of Isabelle/Spartan shows that, despite being built on simple type  
 443 theoretic foundations, Isabelle’s logical framework infrastructure is feasibly able to support  
 444 the development and use of dependent type theory as a proof environment.

445 Many improvements to Spartan are possible: there is currently no general elimination tac-  
 446 tic, definitions for synthesized proof terms should be exported for reusability, the rudimentary  
 447 typechecking mechanism can likely be improved, and automation to enable universe ambiguity  
 448 might prove desirable. Ongoing and future work aims to implement these improvements, as  
 449 well as to establish theoretical properties of the encoding, expand the collection of basic types  
 450 and the library of formalizations, and explore how the framework, along with techniques  
 451 discussed in this paper, may be used to implement related logics like the calculus of inductive  
 452 constructions, two-level type theory [2] and cubical type theory [6].

## 453 Acknowledgements

454 I thank Makarius Wenzel for multiple discussions explaining technical details of the Isabelle  
 455 system, and Cezary Kaliszyk for numerous helpful suggestions for improving this paper.

## 456 References

- 457 1 Dedukti - a Logical Framework, 2020. Accessed 2020-02-13. URL: <https://deducteam.github.io/>.
- 458 2 Danil Annenkov, Paolo Capriotti, Nicolai Kraus, and Christian Sattler. Two-Level Type  
 459 Theory and Applications, 2017. [arXiv:1705.03307](https://arxiv.org/abs/1705.03307).
- 460 3 Ali Assaf, Guillaume Burel, Raphaël Cauderlier, David Delahaye, Gilles Dowek, Catherine  
 461 Dubois, Frédéric Gilbert, Pierre Halmagrand, Olivier Hermant, and Ronan Saillard. Dedukti:  
 462 a Logical Framework based on the  $\lambda\Pi$ -Calculus Modulo Theory. 2020. URL: <http://www.lsv.fr/~dowek/Publi/expressing.pdf>.
- 463 4 Andrej Bauer. Spartan type theory. Talk given at the School and Workshop on Univalent  
 464 Mathematics, University of Birmingham, December 2017. URL: <http://math.andrej.com/2017/12/11/spartan-type-theory/>.
- 465 5 Andrej Bauer, Gaëtan Gilbert, Philipp G. Haselwarter, Anja Petković, Matija Pretnar, and  
 466 Christopher A. Stone. The Andromeda proof assistant. <https://www.andromeda-prover.org/>,  
 467 2020. Accessed 2020-02-13.

- 471   **6**   Cyril Cohen, Thierry Coquand, Simon Huber, and Anders Mörtberg. Cubical Type Theory: a  
472       constructive interpretation of the univalence axiom, 2016. [arXiv:1611.02108](https://arxiv.org/abs/1611.02108).
- 473   **7**   Bart Jacobs and Tom Melham. Translating Dependent Type Theory into Higher Order Logic.  
474       In Marc Bezem and Jan Friso Groote, editors, *Typed Lambda Calculi and Applications*, pages  
475       209–229. Springer Berlin Heidelberg, Berlin, Heidelberg, 1993. doi:[https://doi.org/10.](https://doi.org/10.1007/BFb0037108)  
476       1007/BFb0037108.
- 477   **8**   Lawrence C. Paulson. Isabelle/CTT. Isabelle object logic, bundled with the Isabelle proof  
478       assistant distribution, 1991. URL: [https://isabelle.in.tum.de/website-Isabelle2019/](https://isabelle.in.tum.de/website-Isabelle2019/dist/library/CTT/index.html)  
479       dist/library/CTT/index.html.
- 480   **9**   Lawrence C. Paulson. *Isabelle’s Logics*, June 2019. URL: [https://isabelle.in.tum.de/](https://isabelle.in.tum.de/website-Isabelle2019/dist/Isabelle2019/doc/logics.pdf)  
481       website-Isabelle2019/dist/Isabelle2019/doc/logics.pdf.
- 482   **10**   Frank Pfenning. Elf: A meta-language for deductive systems. In Alan Bundy, editor, *Automated*  
483       *Deduction — CADE-12*, pages 811–815. Springer Berlin Heidelberg, Berlin, Heidelberg, 1994.  
484       doi:[https://doi.org/10.1007/3-540-58156-1\\_66](https://doi.org/10.1007/3-540-58156-1_66).
- 485   **11**   Frank Pfenning. Logical Frameworks. In Alan Robinson and Andrei Voronkov, editors,  
486       *Handbook of Automated Reasoning*, Handbook of Automated Reasoning, chapter 17, pages 1063–  
487       1147. North-Holland, Amsterdam, 2001. doi:[https://doi.org/10.1016/B978-044450813-3/](https://doi.org/10.1016/B978-044450813-3/50019-9)  
488       50019-9.
- 489   **12**   Frank Pfenning and Carsten Schürmann. System Description: Twelf — A Meta-Logical Frame-  
490       work for Deductive Systems. In *Automated Deduction — CADE-16*, pages 202–206. Springer  
491       Berlin Heidelberg, Berlin, Heidelberg, 1999. doi:[https://doi.org/10.1007/3-540-48660-7\\_](https://doi.org/10.1007/3-540-48660-7_14)  
492       14.
- 493   **13**   Univalent Foundations Program. *Homotopy Type Theory: Univalent Foundations of Mathem-*  
494       *atics*. 2013. URL: <https://homotopytypetheory.org/book>.
- 495   **14**   University of Cambridge, Technische Universität München, and contributors. Isabelle2019,  
496       June 2019. URL: <https://isabelle.in.tum.de/website-Isabelle2019/>.
- 497   **15**   Makarius Wenzel. *The Isabelle/Isar Reference Manual*, June 2019. URL: [https://isabelle.](https://isabelle.in.tum.de/website-Isabelle2019/dist/Isabelle2019/doc/isar-ref.pdf)  
498       in.tum.de/website-Isabelle2019/dist/Isabelle2019/doc/isar-ref.pdf.
- 499   **16**   Makarius Wenzel, Lawrence C. Paulson, and Tobias Nipkow. The Isabelle Framework. In  
500       Otmane Ait Mohamed, César Muñoz, and Sofène Tahar, editors, *Theorem Proving in Higher*  
501       *Order Logics*, pages 33–38. Springer Berlin Heidelberg, Berlin, Heidelberg, 2008.
- 502   **17**   Markus Wenzel. Isar - A Generic Interpretative Approach to Readable Formal Proof Documents.  
503       In Yves Bertot, Gilles Dowek, André Hirschowitz, Christine Paulin-Mohring, and Laurent  
504       Théry, editors, *Theorem Proving in Higher Order Logics, 12th International Conference,*  
505       *TPHOLs’99, Nice, France, September, 1999, Proceedings*, volume 1690 of *Lecture Notes in*  
506       *Computer Science*, pages 167–184. Springer, 1999. doi:10.1007/3-540-48256-3\_12.